



راه اندازی ماژول GPS مدل NEO6M قسمت 3 – موقعیت یابی

راه اندازی ماژول GPS مدل NEO6M قسمت 3 موقعیت یابی



<https://blog.microele.com>

تاریخ انتشار ۱۳ آبان، ۱۴۰۰ توسط محمد جواد رشیدیانفر

سلام به همه میکروالکامی ها. در قسمت قبلی از سری آموزش موقعیت یابی و راه اندازی ماژول GPS مدل NEO-6m به تحلیل دقیق و بررسی کد های NMEA خروجی ماژول پرداخته شد. در این قسمت با استفاده از خروجی ماژول و استفاده از کتابخانه TingGPS++ با استفاده از ماژول، موقعیت یابی را انجام داده و موقعیت را بر روی نقشه نمایش خواهیم داد. پس با من تا انتهای مطلب همراه باشید. همچنین شما می توانید سایر مطلب من رو از این قسمت مطالعه کنید.



وسایل مورد نیاز

- ماژول NEO6M یا برد راه انداز آن
- مبدل سریال
- آنتن GPS
- مبدل SMA به UFL
- مبدل ولتاژ DC-DC

معرفی کتابخانه TinyGPS

همانطور که پیشتر و در مطالب قبلی ذکر شد، خروجی ماژول های GPS حاوی **پکت های NMEA** می باشد. برای اینکه از اطلاعات بدست آمده بتوان براحتی استفاده کرد، نیاز به تجزیه و تفکیک آنها داریم. برای اینکار دو راه وجود دارد. راه اول استفاده از کتابخانه های آماده در این خصوص است و راه دیگر، بصورت کدنویسی خود ما این کار را انجام دهیم. راه اول خیلی ساده تر است و شما صرفا توابع و دستورات مربوط به کتابخانه را فراخوانی کرده و نتیجه را دریافت خواهید کرد. اما در راه دوم نیاز به برنامه نویسی دقیق تر است تا نتیجه حاصل شود.

برای تجزیه و استخراج اطلاعات NMEA می توان از کتابخانه TinyGPS++ بهره برد. برای دانلود این کتابخانه جهت استفاده در Arduino IDE از **این لینک** استفاده کنید. همچنین برای فهمیدن نحوه استفاده از کتابخانه، نحوه عملکرد و اطلاعات بیشتر در باره این کتابخانه به **وبسایت ارائه دهنده این کتابخانه** مراجعه نمایید.

برنامه نویسی

در این قسمت به بررسی برنامه و کد های مورد نیاز جهت پیدا کردن موقعیت مکانی پرداخته می شود.



بخش ابتدایی و تعاریف

در ابتدای کد ها، کتابخانه های GPS برای راه اندازی ماژول NEO-6M و سریال نرم افزاری را فراخوانی کرده و در ادامه تنظیمات مربوطه را لحاظ می کنیم.

```
#include <TinyGPS++.h>
#include <SoftwareSerial.h>

int RXPin = 12;
int TXPin = 11;
int GPS_Baudrate = 9600;

// ساخت object از کتابخانه TinyGPS++
TinyGPSPlus gps;

SoftwareSerial gpsSerial(RXPin, TXPin);
```

تابع نمایش اطلاعات displayInfo

در ادامه به هر بخش از این تابع پرداخته خواهد شد.

یافتن و نمایش موقعیت جغرافیایی:

برای تجزیه و یافتن عرض جغرافیایی در بین NMEA دریافتی از ماژول، از دستور `gps.location.lat()` استفاده می شود. جهت بدست آوردن طول جغرافیایی از دستور `gps.location.lng()` استفاده خواهد شد.

```
if (gps.location.isValid())
{
  Serial.print("Latitude: ");
  Serial.println(gps.location.lat(), 6);
  Serial.print("Longitude: ");
```



```
Serial.println(gps.location.lng(), 6);  
Serial.print("Altitude: ");  
Serial.println(gps.altitude.meters());  
}
```

نمایش تاریخ:

جهت نمایش تاریخ میلادی (تاریخ، روز و ماه) از دستورات زیر استفاده می‌شود.

```
if (gps.date.isValid())  
{  
    Serial.print(gps.date.month());  
    Serial.print("/");  
    Serial.print(gps.date.day());  
    Serial.print("/");  
    Serial.println(gps.date.year());  
}
```

نمایش ساعت:

جهت نمایش ساعت (ساعت، دقیقه، ثانیه و صدم ثانیه) از دستورات زیر استفاده می‌شود.

```
if (gps.time.isValid())  
{  
    if (gps.time.hour() < 10) Serial.print(F("0"));  
    Serial.print(gps.time.hour());  
    Serial.print(":");  
    if (gps.time.minute() < 10) Serial.print(F("0"));  
    Serial.print(gps.time.minute());  
    Serial.print(":");  
    if (gps.time.second() < 10) Serial.print(F("0"));  
    Serial.print(gps.time.second());  
}
```



```
Serial.print(".");  
if (gps.time.centisecond() < 10) Serial.print(F("0"));  
Serial.println(gps.time.centisecond());  
}
```

تابع setup

از این تابع برای تنظیمات برنامه در اولین راه اندازی استفاده می شود.

```
void setup()  
{  
  Serial.begin(9600);  
  gpsSerial.begin(GPS_Baudrate);  
}
```

تابع اصلی loop

در تابع اصلی همواره خروجی ماژول توسط پورت سریال نرم افزاری معرفی شده بررسی می شود. در صورتی که دیتا معتبری یافت شود آنها را بروی سریال مانیتور نمایش خواهیم داد. در صورتی که 5 ثانیه بگذرد و دیتا معتبری وجود نداشته باشد عبارت "No GPS detected" نمایش داده خواهد شد.

```
void loop()  
{  
  // نمایش اطلاعات خروجی  
  while (gpsSerial.available() > 0)  
    if (gps.encode(gpsSerial.read()))  
      displayInfo();  
  
  if (millis() > 5000 && gps.charsProcessed() < 10)  
  {  
    Serial.println("No GPS detected");  
  }
```



```
while(true);  
}  
}
```

سورس کد کامل موقعیت یابی با NEO-6M

```
#include <TinyGPS++.h>  
#include <SoftwareSerial.h>  
  
int RXPin = 12;  
int TXPin = 11;  
int GPS_Baudrate = 9600;  
  
// ساخت object از کتابخانه TinyGPS++  
TinyGPSPlus gps;  
  
SoftwareSerial gpsSerial(RXPin, TXPin);  
  
void displayInfo()  
{  
  if (gps.location.isValid())  
  {  
    Serial.print("Latitude: ");  
    Serial.println(gps.location.lat(), 6);  
    Serial.print("Longitude: ");  
    Serial.println(gps.location.lng(), 6);  
    Serial.print("Altitude: ");  
    Serial.println(gps.altitude.meters());  
  }  
  else  
  {
```



```
Serial.println("Location: Not Available");
}
Serial.print("Date: ");
if (gps.date.isValid())
{
    Serial.print(gps.date.month());
    Serial.print("/");
    Serial.print(gps.date.day());
    Serial.print("/");
    Serial.println(gps.date.year());
}
else
{
    Serial.println("Not Available");
}

Serial.print("Time: ");
if (gps.time.isValid())
{
    if (gps.time.hour() < 10) Serial.print(F("0"));
    Serial.print(gps.time.hour());
    Serial.print(":");
    if (gps.time.minute() < 10) Serial.print(F("0"));
    Serial.print(gps.time.minute());
    Serial.print(":");
    if (gps.time.second() < 10) Serial.print(F("0"));
    Serial.print(gps.time.second());
    Serial.print(".");
    if (gps.time.centisecond() < 10) Serial.print(F("0"));
    Serial.println(gps.time.centisecond());
}
else
```



```
{  
  Serial.println("Not Available");  
}  
  
Serial.println();  
Serial.println();  
delay(1000);  
}  
  
void setup()  
{  
  Serial.begin(9600);  
  gpsSerial.begin(GPS_Baudrate);  
}  
  
void loop()  
{  
  // نمایش اطلاعات خروجی  
  while (gpsSerial.available() > 0)  
    if (gps.encode(gpsSerial.read()))  
      displayInfo();  
  
  if (millis() > 5000 && gps.charsProcessed() < 10)  
  {  
    Serial.println("No GPS detected");  
    while(true);  
  }  
}
```

برای دانلود سورس کد از [این لینک](#) استفاده کنید.



مشاهده نتیجه خروجی

ابتدا پایه Rx ماژول را به پایه 11 و پایه Tx ماژول را به پایه 12 آردوینو متصل کنید. همچنین با استفاده از یک مبدل کاهنده خروجی آن را روی 3.3v تنظیم کرده و به ماژول NEO-6M وصل کنید. در نهایت برنامه را آپلود نمایید. در تصویر زیر نتیجه را مشاهده خواهید کرد. در هر ثانیه اطلاعات بر روی سریال مانیتور نمایش پیدا خواهد کرد.
نکته: مبدل کاهنده را به پایه GND آردوینو و ماژول GPS وصل کنید.

```
1 #include <TinyGPS++.h>
2 #include <SoftwareSerial.h>
3
4 int RXPin = 12;
5 int TXPin = 11;
6 int GPS_Baudrate = 9600;
7
8
9 ++TinyGPS از کتابخانه object
10 TinyGPSPlus gps;
11
12 SoftwareSerial gpsSerial(RXP, TXP);
13 void setup()
14 {
15   Serial.begin(9600);
16   gpsSerial.begin(GPS_Baudrate);
17 }
18
19 void loop()
20 {
21   // نمایش اطلاعات خروجی
22   while (gpsSerial.available())
23     if (gps.encode(gpsSerial.read()))
24       displayInfo();
25
26   if (millis() > 5000) {
27     Serial.println("No GPS data received");
28     while(true);
29   }
30 }
```

COM3

```
Latitude: 35.6894
Longitude: 51.3890
Altitude: 0.00
Date: 11/3/2021
Time: 16:06:02.00

Latitude: 35.6894
Longitude: 51.3890
Altitude: 0.00
Date: 11/3/2021
Time: 16:06:02.00

Latitude: 35.6894
Longitude: 51.3890
Altitude: 0.00
Date: 11/3/2021
Time: 16:06:02.00
```

Autoscroll Show timestamp Both NL & CR 9600 baud Clear output

Done uploading
Sketch uses 8880 bytes (28%) of program storage space. Maximum is 30720 bytes.
Global variables use 608 bytes (29%) of dynamic memory, leaving 1440 bytes for local variables. Maximum is 2048 bytes.

بدست آوردن موقعیت مکانی با استفاده از ماژول GPS مدل NEO-6M

ساخت لینک گوگل مپ

برای نمایش موقعیت مستقیماً روی **گوگل مپ**، یا طول و عرض جغرافیایی را در سرچ نقشه گوگل وارد می‌کنیم یا لینک آن را می‌سازیم. برای ساخت لینک گوگل مپ موقعیت فعلی خود، از ساختار زیر استفاده نمایید. بجای Latitude عرض جغرافیایی و بجای Longitude طول جغرافیایی بدست آمده و نمایش داده شده در سریال مانیتور را جایگزین



نمایید.

<http://maps.google.com/maps?q=Latitude,Longitude>

بعنوان مثال لینک زیر، موقعیت مایکروالکام را بر روی نقشه نمایش خواهد داد.

<http://maps.google.com/maps?q=32.6574508,51.6655308>

نتیجه گیری

در این مطلب نحوه تجزیه و تفکیک کد های NMEA خروجی ماژول GPS مدل NEO-6M بررسی شد. همچنین با استفاده از طول و عرض جغرافیایی بدست آمده و وارد کردن آن در گوگل مپ و ساخت لینک آن، موقعیت مکانی بدست آمده نمایش داده شد. در قسمت های بعدی به نحوه ارسال موقعیت مکانی با استفاده از دریافت پیامک پرداخته خواهد شد.

امیدوارم از این آموزش کمال بهره را برده باشید. در صورتی که هرگونه نظر یا سوال داشتید درباره این آموزش لطفاً اون رو در انتهای همین صفحه در قسمت دیدگاه ها قرار بدید. در کوتاه ترین زمان ممکن به اون ها پاسخ خواهم داد. اگر این مطلب براتون مفید بود، اون رو حتماً به اشتراک بگذارید. همینطور میتونید این آموزش را پس از اجرای عملی توی اینستاگرام با هشتگ #microelecom به اشتراک بگذارید و **پیج مایکروالکام** (@microelecom) رو هم منشن کنید.